
In this paper a constructive formalization of quantifier elimination is presented, based on a classical

formalization by Tobias Nipkow. The formalization is implemented and verified in the programming

language/proof assistant Agda. It is shown that, as in the classical case, the ability to eliminate

a single existential quantifier may be generalized to full quantifier elimination and consequently a

decision procedure. The latter is shown to have strong properties under a constructive metatheory,

such as the generation of witnesses and counterexamples. Finally, this is demonstrated on a minimal

theory on the natural numbers.

1 Introduction

1.1 Predicate Logic and Quantifier Elimination

A proposition in predicate logic is formed in one of three ways: from an atom, by linking propositions

together using a connective (such as ∨ or ⇒), or by quantifying a proposition with ∀ or ∃. Neglecting

the internal structure of atoms, it is the third that sets predicate logic apart from propositional logic; the

quantifiers greatly enhance the expressiveness of the language.

A drawback of predicate logic is that the truth of a proposition is no longer easy to determine. With

propositional logic an exhaustive enumeration is possible, but this is not so in predicate logic: to do so

on a proposition such as ∀x.(x 6= x+ 1) would require verifying x 6= x+ 1 for every possible value of x,

which—depending on our choice of domain—could be infinite.

There is not always a way around this; predicate logic is indeed undecidable in the general case.

However, a number of specific theories within predicate logic are in fact decidable—and not simply by

admitting exhaustive enumeration. Rather, decidability is shown through quantifier elimination.

The idea behind quantifier elimination is to devise a method to transform any given proposition into

an equivalent one without quantifiers. The latter can typically be decided very easily, and by virtue of

the equivalence the decision applies to the original proposition as well. This allows any proposition in

the theory to be decided, rendering the theory decidable.

1.2 Classical and Constructive Logic

If quantifier elimination is proven to be possible for a theory, that proof is in turn carried out in another

theory, referred to as the metatheory.

Both the theory and metatheory can vary with respect to being classical or constructive. A classical

(meta)theory, by virtue of the law of excluded middle, allows the use of quantifier dualities, De Morgan’s

laws, proof by contradiction, and other results of classical logic.

In a constructive (meta)theory, however, the above are no longer a priori available. Despite such

constraints, there are several advantages. One of the most significant—beside any philosophical argu-

ments for constructivism—is that a proof of existence requires a witness, an actual value that meets the

specified criteria. For example, a constructive proof of ∃x.x > 2 would necessarily consist of a value

for x, and a proof that it is greater than two. As a consequence, not only does a constructive decision

procedure determine whether a proposition is true or not, but it provides significantly more information

about how.

1.3 Formalization

The difficulty of quantifier elimination depends on the theory in question, but even for simple theories it

is quite high—great care must be taken to ensure that the method is sound. Moreover, applying the pro-

cedure to a complicated proposition is likely impractical for a human (especially if it involves conversion

to disjunctive normal form, which can result in a large growth in the number of terms). These factors

encourage computer formalization of quantifier elimination—both in implementing the procedures, and

verifying that they are correct.

Implementation by itself is conceptually straightforward: propositions are represented by some

datatype, and quantifier elimination as procedure(s) that manipulate objects of that datatype. Verification

makes matters more complicated; the implementation must be accompanied by a proof of its correctness,

which certifies that the quantifier elimination procedure always produces a proposition that is equivalent

to the input (and quantifier-free). To facilitate this, both the implementation and correctness proof are

typically written in a proof assistant.

1.4 History, and Related Work

The technique of quantifier elimination has been used over the last century to prove the decidability of a

number of theories. Notable examples include: real and algebraically closed fields, by Alfred Tarski [4];

several theories on the natural numbers under a constructive metatheory, by Jacques Herbrand [6]; and

Presburger arithmetic (addition on the natural numbers), by Mojżesz Presburger [10].

Following Gödel’s incompleteness theorems [5] came several negative results in decidability. One

such result is the essential undecidability of Robinson arithmetic [9] (addition and multiplication on

the natural numbers), which effectively rules out the possibility of a decision procedure for general

arithmetic.

More recently, several (positive) results have been revisited in the context of computer-verified

proofs. Examples include Tobias Nipkow’s framework for and application of quantifier elimination in

Isabelle [8]; Assia Mahboubi and Cyril Cohen’s Coq formalizations of the decidability of real [2] and

algebraically [7] closed fields; and Guillaume Allais’ constructive proof of the decidability of Presburger

arithmetic in Agda [1].

This paper presents a general framework somewhat similar to that of Nipkow [8], however carried

out in Agda under a constructive metatheory.1 It is applied to one of the theories Herbrand [6] showed

to be decidable, using a technique similar to the one that he presented. It also bears substantial simi-

larity to Allais’s Presburger solver [1], despite being developed largely independently: correspondance

between this author and Allais, during which the latter generously shared the source code of his (at the

time unpublished) work, only occurred after the majority of the work presented in this paper had been

completed. Certain presentational aspects were influenced by Allais’ work, however. More importantly,

several apparent novelties of this work—such as the trick used to obviate the need for prenex form—

in fact appeared earlier in the former. The primary contribution of this work is therefore its relative

generality.

1And, alas, without reflection for the time being.

1.5 Organization

The remainder of this paper is organized as follows: First, brief background information is given on

several theoretical aspects (Section 2). Next, a theory-independent formalization of quantifier elimination

is shown (Section 3), followed by an application to a theory on the natural numbers (Sections 4 and 5).

Finally, possibilities for further development are discussed (Section 6).

The source code for the project (excluding the Agda standard library) is available on GitHub.2 At the

time of this paper’s writing, the code compiles with Agda version 2.5.2, and version 0.13 of the standard

library.

2 Theoretical Background

2.1 Quantifier Elimination

Rather than attempting to remove all quantifiers at once, an incremental approach is usually taken, dra-

matically reducing the scope of the problem. A procedure is devised to remove a single quantifier, often

∃, from an otherwise quantifier-free proposition:

∃x.φ ⇐⇒ ψ ,

where φ and ψ are quantifier-free. Using the quantifier duality ∀x.φ ⇐⇒ ¬∃x.¬φ (in a classical theory)

this can be adapted to remove ∀ as well. If the full proposition in question (which may contain many

quantifiers) is placed into prenex form, where all of its quantifiers are pushed as far out as possible, then

repeated application of the single-step procedure can clearly be used to eliminate all quantifiers from the

“inside out”:

∃z.∀y.∃x.φ ⇐⇒ ∃z.∀y.ρ ⇐⇒ ∃z.σ ⇐⇒ ψ ,

noting that φ , ρ , σ , and ψ are all quantifier-free. The same recursive strategy can just as well be used

without placing the proposition into prenex form.3

To narrow the problem even further, the quantifier-free sub-proposition φ can be placed into disjunc-

tive normal form (DNF):

φ ⇐⇒ C1 ∨C2 ∨ . . .∨Cn

where each Ci is a conjunction of literals (a literal being an atomic formula or its negation). This is useful

because existential quantification distributes across disjunction:

∃x.φ ⇐⇒ ∃x.(C1 ∨C2 ∨ . . .∨Cn) ⇐⇒ (∃x.C1)∨ (∃x.C2)∨ . . .∨ (∃x.Cn).

As a result, elimination can be carried out on each conjunction separately, reducing the problem to

quantifier elimination on conjunctions of literals.

Once a quantifier elimination procedure has been shown, decidability of the theory is obtained—

provided that all quantifier-free propositions are decidable. The latter requirement is trivially true for

theories where atomic formulae represent decidable relations (e.g. equality on the natural numbers).

2https://github.com/guspopje/agda-qelim
3This is actually quite important; transforming a proposition into prenex form uses quantifier dualities that are not valid in

constructive logic.

2.2 Agda

Agda [13], the programming language/proof assistant used in this paper, is based on intuitionistic type

theory. As a result it is constructive, with the consequences described in Section 1.2. Further information

about the language is available from Agda’s website and the various tutorials listed there [14].

2.3 Theory, Metatheory, and Semantics

In quantifier elimination, and proofs about logic systems in general, there are frequently two layers: the

theory T under consideration, expressed in the object language, and the metatheory M in which T is

analyzed, expressed in the metalanguage.

The notions of equivalance and decidability (as related to quantifier elimination) necessitate that a

notion of provability or truth be associated with T . One option is to define a proof system directly for

T , as in Herbrand’s thesis [6]. This allows a syntactic treatment, notions such as “equivalent in T ” and

“provable in T ”, and consequently strong separation between theory and metatheory.4

Another option, as taken by Tarski [11], Nipkow [8], and this project, is to instead define the seman-

tics of propositions of T , as propositions in M:

[[ · ]] : T → M.

This is typically accomplished recursively, mapping each each connective or quantifier in T to the corre-

sponding one in M. In the case of this paper, T is an arbitrary theory of first order logic (subject to minor

constraints), and M is the flavor of Intuitionistic Type Theory used by Agda. A notable consequence of M
being constructive (and the definition of [[ · ]] used in this paper) is that the semantics of T are constructive

as well.

With this approach, quantifier elimination produces a proposition that is semantically equivalent to

the original, and in the end it is the semantics of T that are proven to be decidable (as opposed to T itself,

which is not possible without a proof system of its own). As the semantics lie in M, this means that

decidability is shown for a fragment of M.

2.4 De Bruijn Indices

One of the difficulties in formalizing a theory is the handling of free and bound variables. With a tra-

ditional “named variable” approach, extra conditions must be added to prevent substitutions that would

capture free variables.5

An alternative is to use De Bruijn indices, where each occurrence of a bound variable is denoted

instead by a number that indicates how many variable-binders “deep” the occurrence is from the binder

to which it refers. For example, the proposition

∀x.(x ≤ 4∨ (∃y.x = y+5))

is represented as

∀.( 0 ≤ 4∨ (∃. 1 = 0 +5)).

Here it is noted that within the scope of the ∃, the index 0 refers to the ∃ (i.e., the variable y), and 1

refers to the quantifier one layer out, namely ∀ (i.e., the variable x).

4In Herbrand’s case, this allows the analysis of a classical theory under a constructive metatheory.
5An example from lambda calculus is the (invalid) beta reduction of (λx.λy.x)y to λy.y.



6 Formalizing Constructive Quantifier Elimination in Agda

Free variables can be treated the same way; their indices simply point outside of the visible formula

(for example, the rightmost variable in ∃.( 0 ≥ 3 )). The benefits include dramatically simpler rules for

substitution, alpha equivalence for free, and ease in associating values with free variables.

With named variables, the latter (referred to as an environment) is accomplished via a mapping from

names to values, while with de Bruijn indices only a list of values is required—the list corresponds

to the “missing” layers of quantifiers to which the free variables refer. For example, the proposition

∀.∃.( 4 ≥ 7 ) requires a list with at least six values.

In this paper, such a bound is referred to as the arity of a proposition, arising from the interpretation

of a proposition as a function of its free variables.

3 Theory-Independent Work

3.1 Atoms

In the interest of generality, atomic formulae are not represented by a fixed type, but by a type given as a

module parameter. The type is indexed by a natural numbers n representing its arity (an upper bound on

its free variables, as discussed in Section 2.4):

Atom : N→ Set

The internal structure of an Atom is completely unspecified.

The semantics of Atom is also given by way of module parameters. First, the set of values which

variables may take:

Y : Set

Then, a function which gives the semantics for an atom:

[[_]]a : {n : N} → Atom n → Vec Y n → Set

The implicit parameter n : N is the arity of the atom, and the following parameter of type Atom n is the

atom itself. The last parameter, of type Vec Y n, is the environment: a list (vector) of length n of values

for the free variables in the atom (see Section 2.4). This, in a sense, forces Atom to use de Bruijn indices

internally—no names are associated with the values in the enviroment. Moreover, since the enviroment

for an Atom n is a list of n values, the effective arity of the atom is restricted to n, as intended.

Additionally, it is required that the semantics of Atom be decidable under any given enviroment.

This is often the case (as discussed in Section 2.1, and is equivalent to the semantics of all quantifier-free

propositions being decidable. As it turns out, for a constructive theory this is important not only for

decidability but for quantifier elimination itself, as will be seen later on. Another module parameter is

used to implement this requirement:

[[_]]a? : {n : N} (a : Atom n) (e : Vec Y n) → Dec ([[ a ]]a e)

The Dec type family, from Agda’s standard library, is indexed by a type (A). An object of type Dec A is a

decision for A: either a proof that A is inhabited (yes a, where a : A), or a proof that it is not (no x, where

x : ¬ A, i.e. x : A →⊥).

For organizational purposes the above are grouped into a record type, forming an abstract represen-

tation of atoms with decidable semantics:

record DecAtom : Set1 where
field

Atom : N → Set
Y : Set
[[_]]a : {n : N} → Atom n → Vec Y n → Set
[[_]]a? : {n : N} (a : Atom n) (e : Vec Y n) → Dec ([[ a ]]a e)

A single module parameter of type DecAtom is used in lieu of four separate parameters.

3.2 Representation of Propositions

Propositions are represented by following datatype Prop. Its constructors allow the formation of a propo-

sition from an atom, or from other propositions by way of the typical connectives and quantifiers.

data Prop (n : N) : Set where
atom : Atom n → Prop n
⊥⊥ : Prop n
_∨_ : Prop n → Prop n → Prop n
_∧_ : Prop n → Prop n → Prop n
_⇒_ : Prop n → Prop n → Prop n
E_ : Prop (suc n) → Prop n
A_ : Prop (suc n) → Prop n

Negation is defined for convenience:

~_ : {n : N} → Prop n → Prop n
~ φ = φ ⇒⊥⊥

It is noted that because the semantics of a proposition is not a priori decidable, under a constructive

(meta)theory propositions cannot be reduced to a more minimal set of connectives/quantifiers, as would

be typical in a classical setting.

The quantifiers E_ and A_ reflect the use of de Bruijn indices (Section 2.4): neither constructor

accepts any indication of which variable is to be quantified (recall that with de Bruijn indices this is not

needed), and both decrement the arity (by virtue of binding one of the free variables in the quantified

proposition).

3.3 Semantics of Propositions

The semantics of a proposition is then defined recursively from [[_]]a:

[[_]] : {n : N} → Prop n → Vec Y n → Set
[[ ⊥⊥ ]] ys = ⊥

[[ atom a ]] ys = [[ a ]]a ys
[[ φ1 ∨ φ2 ]] ys = ([[ φ1 ]] ys) ⊎ ([[ φ2 ]] ys)
[[ φ1 ∧ φ2 ]] ys = ([[ φ1 ]] ys) × ([[ φ2 ]] ys)
[[ φ1 ⇒ φ2 ]] ys = ([[ φ1 ]] ys) → ([[ φ2 ]] ys)

[[ E φ ]] ys = Σ Y (λ y → [[ φ ]] (y :: ys))
[[ A φ ]] ys = (y : Y) → ([[ φ ]] (y :: ys))

Absurdity, disjunction, conjunction, and implication are respectively mapped to the empty, disjoint

union, cartesian product, and function types.

The semantics of existential quantification is represented using a Σ (dependent sum/pair) type. Mem-

bers of the resulting type are pairs consisting of a value y : Y and an element of the inner proposition’s

semantics with y prepended to the environment, i.e., proof that the inner proposition is true with the first

free variable “set to y”.

The semantics of universal quantification is defined similarly, but using a (dependent) function type6

in place of the Σ type—all values for y must result in the inner proposition being true.

3.4 Quantifier-Free Propositions

As quantifier-free propositions are of importance, a representation of this quality is defined:

data QFree {n : N} : Prop n → Set where
⊥⊥ : QFree ⊥⊥

atom : (a : Atom n) → QFree (atom a)
_∨_ : {φ1 φ2 : Prop n} → QFree φ1 → QFree φ2 → QFree (φ1 ∨ φ2)
_∧_ : {φ1 φ2 : Prop n} → QFree φ1 → QFree φ2 → QFree (φ1 ∧ φ2)
_⇒_ : {φ1 φ2 : Prop n} → QFree φ1 → QFree φ2 → QFree (φ1 ⇒ φ2)

~-qf_ : {n : N} {φ : Prop n} → QFree φ → QFree (~ φ)
~-qf qf = qf ⇒ ⊥⊥

QFree φ is inhabited if and only if φ is quantifier-free.

Semantically speaking, all of the connectives preserve decidability: the result of joining two seman-

tically decidable propositions with _∨_, _∧_, or _⇒_ is also semantically decidable. This is shown

for _⇒_ (with semantics →) as follows:

_→?_ : {A B : Set} → Dec A → Dec B → Dec (A → B)
_ →? (yes b) = yes (λ _ → b)
(yes a) →? (no ¬b) = no (λ f → ¬b (f a))
(no ¬a) →? (no ¬b) = yes (λ a → contradiction a ¬a)

The same property can be shown for _∨_ and _∧_ (with semantics _⊎_ and _×_) in a similar

manner, resulting in the following two functions:

_×?_ : {A B : Set} → Dec A → Dec B → Dec (A × B)
_⊎?_ : {A B : Set} → Dec A → Dec B → Dec (A ⊎ B)

It is also noted that the semantics of ⊥⊥, namely ⊥, is trivially decidable.

Given the above and that the semantics for atoms are decidable ([[_]]a?), it follows by induction that

the semantics of any quantifier-free proposition is decidable:

qfree-dec : {n : N} → (φ : Prop n) → QFree φ → (e : Vec Y n) → Dec ([[ φ ]] e)

6A Π type, though Agda’s syntax makes it of little use to write it as such.

3.5 Quantifier Elimination

As discussed in Section 2.1, quantifier elimination is typically accomplished by eliminating existential

quantifiers one by one, from the “inside out”. It is performed in that order so that when a quantifier is be-

ing eliminated, the enclosed proposition is already quantifier-free, simplifying the problem significantly.

The method by which a single quantifier is eliminated depends on the theory under consideration,

making it impossible to directly define (whilst maintaining generality). Instead—in a similar manner to

DecAtom—it is defined abstractly with a record type QE which captures the necessary properties of a

single-step elimination procedure. A specific implementation takes the form of an object qe : QE.

record QE : Set where
field

elim : {n : N} (φ : Prop (suc n)) → QFree φ → Prop n
qfree : {n : N} (φ : Prop (suc n)) (qf : QFree φ) → QFree (elim φ qf)
equiv : {n : N} (φ : Prop (suc n)) (qf : QFree φ) (e : Vec Y n) →
[[ E φ ]] e ↔ [[ elim φ qf ]] e

The field elim represents the single-step elimination procedure itself, accepting a quantifier-free

proposition with up to n+ 1 free variables and producing one with up to n. It is noted that the input

to elim does not contain the existential quantifier to eliminate, rather it is implied—for example, to elim-

inate the quantifier from E φ , the elim procedure is invoked on just φ . The field qfree represents a proof

that elim always produces a quantifier-free proposition. Finally, equiv establishes elim’s correctness—

that the propositions E φ and elim φ ... are semantically equivalent.7

Such a single-step procedure can then be “lifted” to eliminate all quantifiers from a proposition

via recursion on the proposition’s structure (the general approach, as stated before, being to eliminate

quantifiers from the inside out). The cases are as follows:

1. The absurd proposition (⊥⊥); it is left unchanged.

2. An atom; it is left unchanged.

3. A proposition formed from disjunction, conjunction, or implication (∨, ∧, or ⇒); the sub-proposition(s)

are quantifier-eliminated recursively.

4. An existentially-quantified proposition (E φ ); φ is quantifier-eliminated recursively, and elim is

applied to the result.

5. A universally-quantified proposition (A φ ); φ is quantifier-eliminated recursively, and the quanti-

fier is treated as its (classical) existential dual (~ E ~).8

This procedure is formalized as the function lift-qe:

lift-qe : {n : N} → QE → Prop n → Prop n
lift-qe-qfree : {n : N} (qe : QE) (φ : Prop n) → QFree (lift-qe qe φ)

lift-qe _ ⊥⊥ = ⊥⊥

lift-qe _ (atom a) = atom a
lift-qe qe (φ1 ∨ φ2) = (lift-qe qe φ1) ∨ (lift-qe qe φ2)

7The notation A ↔ B is defined as ( A → B ) × ( B → A ).
8The validity of this under a constructive metatheory is not immediately obvious, and will be addressed Section 3.5.1.

lift-qe qe (φ1 ∧ φ2) = (lift-qe qe φ1) ∧ (lift-qe qe φ2)
lift-qe qe (φ1 ⇒ φ2) = (lift-qe qe φ1) ⇒ (lift-qe qe φ2)
lift-qe qe (E φ) = QE.elim qe (lift-qe qe φ) (lift-qe-qfree qe φ)
lift-qe qe (A φ) = ~ (QE.elim qe (~ lift-qe qe φ) (~-qf lift-qe-qfree qe φ))

The function lift-qe-qfree (contents omitted) affirms that lift-qe does indeed eliminate quantifiers, via

recursion on the proposition’s structure and the use of QE.qfree.

3.5.1 Correctness

The correctness of the lifted procedure—that lift-qe qe φ is equivalent to φ—is proven recursively based

on the correctess of the single-step procedure. This takes the form of two functions, proving each direc-

tion of the equivalence:

lift-qe-fwd : {n : N} (qe : QE) (φ : Prop n) (e : Vec Y n) → [[ φ ]] e → [[ lift-qe qe φ ]] e
lift-qe-bwd : {n : N} (qe : QE) (φ : Prop n) (e : Vec Y n) → [[ lift-qe qe φ ]] e → [[ φ ]] e

For both directions, the cases ⊥⊥ and atom are trivial; the former is impossible and the latter is un-

changed by lift-qe. For ∨, ∧, and ⇒, correctness of lift-qe is proven recursively on each sub-proposition,

and then combined:

lift-qe-fwd qe (φ1 ∨ φ2) e = Sum.map (lift-qe-fwd qe φ1 e) (lift-qe-fwd qe φ2 e)
lift-qe-fwd qe (φ1 ∧ φ2) e = Product.map (lift-qe-fwd qe φ1 e) (lift-qe-fwd qe φ2 e)
lift-qe-fwd qe (φ1 ⇒ φ2) e = λ f → lift-qe-fwd qe φ2 e ◦ f ◦ lift-qe-bwd qe φ1 e

lift-qe-bwd qe (φ1 ∨ φ2) e = Sum.map (lift-qe-bwd qe φ1 e) (lift-qe-bwd qe φ2 e)
lift-qe-bwd qe (φ1 ∧ φ2) e = Product.map (lift-qe-bwd qe φ1 e) (lift-qe-bwd qe φ2 e)
lift-qe-bwd qe (φ1 ⇒ φ2) e = λ f → lift-qe-bwd qe φ2 e ◦ f ◦ lift-qe-fwd qe φ1 e

In the case of existential quantification, lift-qe recurses on φ , producing an equivalent, quantifier-free

ψ , which QE.elim is applied to. The reasoning behind this is as follows:

∃x.φ ⇐⇒ ∃x.ψ ⇐⇒ elim(ψ).

The first equivalence is justified by the correctness of lift-qe on φ , obtained recursively, and the

second by the correctness of the the single-step procedure, given by QE.equiv. Formalized:

lift-qe-fwd qe (E φ) e
= proj1 (QE.equiv qe (lift-qe qe φ) (lift-qe-qfree qe φ) e)
◦ Σ-map (λ y → lift-qe-fwd qe φ (y :: e))

lift-qe-bwd qe (E φ) e
= Σ-map (λ y → lift-qe-bwd qe φ (y :: e))
◦ proj2 (QE.equiv qe (lift-qe qe φ) (lift-qe-qfree qe φ) e)

where Σ-map proves that if ∀x.(B(x) ⇒ C(x)), then ∃x.B(x) ⇒ ∃x.C(x), in this case used to obtain

∃x.φ ⇐⇒ ∃x.ψ from φ ⇐⇒ ψ :

Σ-map : {A : Set} {B C : A → Set} → ((a : A) → B a → C a) → Σ A B → Σ A C
Σ-map f (a , b) = (a , f a b)

The case of universal quantification is cause for mild concern, however: lift-qe treats the quantifier

A as its classical dual ~ E ~.

In a classical metatheory, correctness could be obtained as follows (once again taking ψ to be the

quantifier-free equivalent of φ ):

∀x.φ ⇐⇒ ¬∃x.¬φ ⇐⇒ ¬∃x.¬ψ ⇐⇒ ¬elim(¬ψ).

The first equivalence is justified by quantifier duality, the second by the correctness of lift-qe on φ (φ ⇐⇒

ψ , obtainable via recursion), and the third by the correctness of QE.elim (QE.equiv). Conceptually, this

corresponds to treating ∀ as ¬∃¬ from the outset.

Under a constructive metatheory, though, the first equivalence is not valid due to the lack of complete

quantifier duality: while ∀x.φ ⇒¬∃x.¬φ , the converse is not provable. However, in this case this can be

neatly sidestepped by rearranging things slightly:

∀x.φ ⇐⇒ ∀x.ψ ⇐⇒ ¬∃x.¬ψ ⇐⇒ ¬elim(¬ψ).

The difference here is that the quantifier duality is applied to ψ , instead of φ . ψ , being quantifier-free,

has decidable semantics (by qfree-dec), and as a consequence the necessary quantifier duality can in fact

be proven. General forms of the duality are formalized as follows:

∀-duality-fwd : {A : Set} {B : A → Set} → ((a : A) → B a) →¬ Σ A (¬_ ◦ B)
∀-duality-fwd all-true (a , is-false) = is-false (all-true a)

∀-duality-bwd : {A : Set} {B : A → Set} → ((a : A) → Dec (B a)) →
¬ Σ A (¬_ ◦ B) → ((a : A) → B a)

∀-duality-bwd decide none-false a with decide a
... | yes a-true = a-true
... | no a-false = ⊥-elim (none-false (a , a-false))

It is noted that the “backward” direction requires that B be decidable.9 The correctness proof then

proceeds as outlined above:

lift-qe-fwd qe (A φ) e
= contraposition (proj2 (QE.equiv qe (~ lift-qe qe φ) (~-qf lift-qe-qfree qe φ) e))
◦ ∀-duality-fwd
◦ Π-map (λ y → lift-qe-fwd qe φ (y :: e))

lift-qe-bwd qe (A φ) e
= Π-map (λ y → lift-qe-bwd qe φ (y :: e))
◦ ∀-duality-bwd (λ y → qfree-dec (lift-qe qe φ) (lift-qe-qfree qe φ) (y :: e))
◦ contraposition (proj1 (QE.equiv qe (~ lift-qe qe φ) (~-qf lift-qe-qfree qe φ) e))

where Π-map is the dependent product/universal quantification counterpart of Σ-map:

9While it could have been formulated to use the weaker requirement that ¬ ¬ B a → B a, there is no particular benefit to

doing so in this case.

Π-map : {A : Set} {B C : A → Set} →

((a : A) → B a → C a) → ((a : A) → B a) → ((a : A) → C a)
Π-map f g a = f a (g a)

3.6 Decidability

Given a single-step elimination procedure qe : QE, the decidability of any proposition φ follows: lift-qe
qe φ produces an equivalent, quantifier-free proposition ψ . As such, ψ is decidable (qfree-dec). Because

φ and ψ are semantically equivalent (lift-qe-fwd, lift-qe-bwd), this immediately results in the decidability

of φ .

[[_]]? : {n : N} → (φ : Prop n) → (e : Vec Y n) → Dec ([[ φ ]] e)
[[ φ ]]? e with qfree-dec (lift-qe qe φ) (lift-qe-qfree qe φ) e
... | yes [[ψ ]] = yes (lift-qe-bwd qe φ e [[ψ ]])
... | no ¬[[ψ ]] = no (¬[[ψ ]] ◦ lift-qe-fwd qe φ e)

The theory in question, whatever it may be, is thus proven decidable.

3.6.1 Consequences

With decidability, the law of excluded middle can be proven for the semantics of Prop:

LEM : {n : N} (φ : Prop n) (e : Vec Y n) → [[ φ ∨ (~ φ) ]] e
LEM φ e with [[ φ ]]? e
... | yes [[φ ]] = inj1 [[φ ]]
... | no ¬[[φ ]] = inj2 ¬[[φ ]]

Using the law of excluded middle, the hitherto unavailable classical results become provable—with

all of the benefits of a constructive metatheory (see Section 1.2). For example, for any proposition φ ,

(∀x.φ)∨ (∃x.¬φ) is provable:

∀-or-∃¬ : {n : N} (φ : Prop (suc n)) (e : Vec Y n) → [[ (A φ) ∨ (E ~ φ) ]] e
∀-or-∃¬ φ e with [[ E (~ φ) ]]? e
... | yes [[E~φ ]] = inj2 [[E~φ ]]
... | no ¬[[E~φ ]]

= inj1 (λ y → [ id , (λ ¬[[φ ]] → contradiction (y , ¬[[φ ]]) ¬[[E~φ ]]) ]′ (LEM φ (y :: e)))

While trivial in a classical setting, in a constructive setting this produces either (i) a proof that φ is true

for every x, or (ii) a counterexample—a value for x which causes φ to be false (and a proof thereof).

Analogous results are obtained for the theorem (∃x.φ)∨ (∀x.¬φ).

3.7 Disjunctive Normal Form and Products

The most basic formulation of a single-step elimination procedure (QE) is one that accepts a quantifier-

free proposition φ , and produces a quantifier-free proposition ψ such that (∃x.φ) ⇐⇒ ψ . There are no

restrictions on the form of φ , other than that it is quantifier-free. In practice, many quantifier elimination

procedures require that φ be transformed into a special form first, as seen in the works of Herbrand [6],

Nipkow [8], and Allais [1].

As discussed in Section 2.1 one such form is Disjunctive Normal Form (DNF), where propositions

take the shape of a disjunction of conjunctions of literals. The utility of this form is that since existential

quantification distributes across disjunction, the problem of quantifier elimination on DNF formulae

trivially reduces to quantifier elimination on conjunctions of literals:

∃x.φ ⇐⇒ ∃x.(C1 ∨C2 ∨ . . .∨Cn) ⇐⇒ (∃x.C1)∨ (∃x.C2)∨ . . .∨ (∃x.Cn).

The mechanics and correctness of the conversion to DNF are not discussed here, as they do not differ

significantly from under a classical metatheory.10 The results are a function dnf that transforms any

proposition into DNF, and proof of its correctness:

dnf : {n : N} (p : Prop n) (qf : QFree p) → DNF n
dnf-fwd : {n : N} (p : Prop n) (qf : QFree p) (e : Vec Y n) → [[ p ]] e → [[ D.i (dnf p qf) ]] e
dnf-bwd : {n : N} (p : Prop n) (qf : QFree p) (e : Vec Y n) → [[ D.i (dnf p qf) ]] e → [[ p ]] e

The DNF datatype employed above is a list of lists of literals, which can be “interpreted” as a propo-

sition using the function D.i.11

Single-step elimination on propositions in DNF is defined much like the more general QE:

record DNFQE : Set where
field

elim : {n : N} → DNF (suc n) → Prop n
qfree : {n : N} (φ : DNF (suc n)) → QFree (elim φ)
equiv : {n : N} (φ : DNF (suc n)) (e : Vec Y n) → [[ E (D.i φ) ]] e ↔ [[ elim φ ]] e

Single-step elimination on conjunctions of literals (referred to here as products) is also defined:

record ProdQE : Set where
field

elim : {n : N} → Prod (suc n) → Prop n
qfree : {n : N} (φ : Prod (suc n)) → QFree (elim φ)
equiv : {n : N} (φ : Prod (suc n)) (e : Vec Y n) → [[ E (P.i φ) ]] e ↔ [[ elim φ ]] e

The Prod datatype is shorthand for a list of literals, and, analogously to DNF, is interpreted as a

proposition using P.i.
The distribution of ∃ across disjunctions allows single-step elimination on products to be trivially

generalized to single-step elimination on propositions in DNF (proof omitted):

Prod⇒DNF.lift : ProdQE → DNFQE

The conversion of any quantifier-free proposition to DNF (dnf, dnf-fwd, dnf-bwd above) allows

single-step elimination on propositions in DNF to be generalized to single-step elimination on any

quantifier-free proposition (proof omitted):

lift-dnf-qe : DNFQE → QE

10Recall that the semantics of an atom is decidable, so De Morgan’s laws hold.
11An alternative approach is to use an actual Prop, along with a proof that it is in disjunctive normal form, but this can make

manipulation cumbersome.

Finally, composing these two functions allows single-step elimination on products to be generalized

to single-step elimination on any quantifier-free proposition:

lift-prod-qe : ProdQE → QE
lift-prod-qe = lift-dnf-qe ◦ Prod⇒DNF.lift

The scope of the problem is therefore narrowed considerably: to obtain full quantifier elimination

and decidability (recalling the results of Sections 3.5 and 3.6, lift-qe and [[_]]? in particular), one need

only create a single-step quantifier elimination procedure that acts on products. This can be put to direct

use on a number of theories, as described by Nipkow [8]. The following section gives an overview of

one such application.

4 The Theory of Successor

The example theory to which this framework is applied is the theory of successor on the natural numbers

(“SN”), as presented by Herbrand [6]. Atomic formulae are equalities between terms, each of which is

the application of the successor function S some (known) number of times to either a variable or zero.

For example:

S(S(S(x))) = S(S(S(S(0))))

or, more concisely:

S3(x) = S4(0).

For convenience the previous atomic formula may be written as x+ 3 = 4, but it is important to note

that—unlike in the case of Presburger arithmetic—one may not add variables together, as this would

represent an unknown number of applications of S. Thus x+ 3 = y+ 7 corresponds to a valid atomic

formula, while x+ y = z+5 does not.

The machinery developed in Section 3 allows the scope of quantifier elimination to be reduced to

elimination of a single variable x from a conjunction of literals (equalities or negated equalities, i.e.

inequalities). For this theory, that can be accomplished via substitution; if an equality involving x is

found, such as x+5= y+3, then substitutions are made accordingly throughout the conjunction, thereby

removing x, and the inequalities y 6= 0 and y 6= 1 are added. On the other hand, if x only occurs in

inequalities, then those inequalities may be dropped from the conjunction—each one is only false for at

most one value of x, so there is always some value for x which satisfies them all.12

The (rather verbose) formalization of the above, omitted from this paper in the interest of brevity,

results in a ProdQE object. This is then lifted to a QE object via lift-prod-qe, leading to full quantifier

elimination via lift-qe and decidability via [[_]]?.

5 Demonstration

The resulting decision procedure and consequences are demonstrated on several small propositions in

order to give a sense of the benefits offered by a constructive approach. The syntax for SN propositions

leaves much to be desired; what they code for is indicated with comments. First, a simple system of

equalities:

12This excludes trivial inequalities such as x+3 6= x+3 and x+2 6= x+4, which are simplified instead of dropped.

test0 : Prop zero
test0 = E E ( – ∃x.∃y.

(atom (S 3 (var (fsuc fzero)) == S 1 (var fzero))) – 3 + x = 1 + y

∧ (atom (S 8 ∅ == S 4 (var fzero))) – 8 = 4 + y

)

Normalizing [[ test0 ]]? [] yields:13

yes (2,4,refl,refl)

The 2 and 4 are witnesses to the existential quantifiers, which is to say values for x and y, and the pair of

refl constitute a proof of the inner conjunction (under the environment [4,2]).

Next, a proposition with a universal quantifier is decided:

– ∀x.(x=0 ∨ ∃y.x=y+1)

test1 : Prop zero
test1 = A ((atom (S 0 (var fzero) == S 0 ∅))
∨ (E (atom (S 0 (var (fsuc fzero)) == S 1 (var fzero)))))

[[ test1 ]]? [] normalizes to yes followed by a (424-line) function that proves the inner proposition for any

given x.

Finally, a proposition with a free variable is examined:

– (x = 0) ∨ (∃y.x=y+2)

test2 : Prop 1
test2 = ((atom (S 0 (var fzero) == S 0 ∅))
∨ (E (atom (S 0 (var (fsuc fzero)) == S 2 (var fzero)))))

The function ∀-or-∃¬ (Section 3.6) is run on test2. As test2 is not true for all values, a counterexample is

produced instead: inj2 ( 1 , ... ), 1 being a value for which the proposition does not hold, and ... consisting

of a trivial proof that 1 6= 0 and a lengthy proof that no y exists such that 1 = 2+ y.

6 Future Work

While the code for the “core” of the theory-independent portion is relatively well-organized, the same

cannot be said for the DNF-conversion portion. The handling of trivially true or false factors in the latter

is also suboptimal: those that are trivially true can (and should) be removed from products, as they have

no effect other than fueling DNF explosion, and those that are false imply that the product as a whole is

false, and therefore equivalent to ⊥ (the recognition of which would greatly speed up the procedure, as

well as contributing less to the aforementioned DNF explosion).

Another improvement would be the application of the framework to more expressive theories, such

as Presburger arithmetic or any of the other theories discussed in Section 1.4. In the former case, it may

be possible to adapt Allais’s proof [1].

A final area of future work noted here is the implemention of reflection, where propositions in the

metatheory can be manipulated directly (or, seemingly so). This would allow quantifier elimination to

13Technically, each refl appeared as .Agda.Builtin.Equality._≡_.refl.

be applied to suitable Agda propositions directly, without the need to first encode them as Prop. Such a

strategy is employed in Nipkow’s framework [8] in Isabelle, and the Omega solver [3] provides similar

functionality for Presburger arithmetic in Coq (albeit as a tactic, rather than through reflection). Agda’s

reflection mechanism is discussed in a more general context by van der Walt [12].

